# $Id: PKGBUILD 175183 2013-01-16 10:19:35Z jgc $
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=xorg-bdftopcf
pkgver=1.0.4
pkgrel=1
pkgdesc="Convert X font from Bitmap Distribution Format to Portable Compiled Format"
arch=(i686 x86_64)
url="http://xorg.freedesktop.org/"
license=('custom')
depends=('libxfont')
makedepends=('xorg-util-macros' 'xproto')
groups=('xorg' 'xorg-apps')
source=(http://xorg.freedesktop.org/archive/individual/app/bdftopcf-${pkgver}.tar.bz2)
sha256sums=('eaf59057ba3d7cffe29526562ce50868da7da823487a4cfb3e16946e5ffd2798')

build() {
  cd "${srcdir}/bdftopcf-${pkgver}"
  ./configure --prefix=/usr
  make
}

package() {
  cd "${srcdir}/bdftopcf-${pkgver}"
  make DESTDIR="${pkgdir}" install

  install -Dm644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/COPYING"
}
